

Syracuse University

SURFACE

Electrical Engineering and Computer Science -
Technical Reports

College of Engineering and Computer Science

9-1992

Multiple-Length Division Revisited: A Tour of the Minefield

Per Brinch Hansen

Syracuse University, School of Computer and Information Science, pbh@top.cis.syr.edu

Follow this and additional works at: https://surface.syr.edu/eecs_techreports



Part of the [Computer Sciences Commons](#)

Recommended Citation

Hansen, Per Brinch, "Multiple-Length Division Revisited: A Tour of the Minefield" (1992). *Electrical Engineering and Computer Science - Technical Reports*. 166.

https://surface.syr.edu/eecs_techreports/166

This Report is brought to you for free and open access by the College of Engineering and Computer Science at SURFACE. It has been accepted for inclusion in Electrical Engineering and Computer Science - Technical Reports by an authorized administrator of SURFACE. For more information, please contact surface@syr.edu.

SU-CIS-92-19

***Multiple-Length Division Revisited:
A Tour of the Minefield***

Per Brinch Hansen

September 1992

*School of Computer and Information Science
Syracuse University
Suite 4-116, Center for Science and Technology
Syracuse, NY 13244-4100*

Multiple-Length Division Revisited: A Tour of the Minefield¹

PER BRINCH HANSEN

Syracuse University, Syracuse, New York 13244, U.S.A.

September 1992

SUMMARY

Long division of natural numbers plays a crucial role in Cobol arithmetic, cryptography, and primality testing. Only a handful of textbooks discuss the theory and practice of long division, and none of them do it satisfactorily. This tutorial attempts to fill this surprising gap in the literature on computer algorithms. We illustrate the subtleties of long division by examples, define the problem concisely, summarize the theory, and develop a complete Pascal algorithm using a consistent terminology.

KEY WORDS Algorithms Multiple-length integer division

INTRODUCTION

Long division of natural numbers plays a crucial role in *Cobol arithmetic* [1], *cryptography* [2], and *primality testing* [3]. While writing a program for primality testing [4], the author learned two lessons the hard way:

1) Only a handful of textbooks discuss the theory and practice of long division, and none of them do it satisfactorily.

2) A correct, efficient algorithm for long division cannot be reinvented with minimal effort.

This tutorial attempts to fill this surprising gap in the literature on computer algorithms.

A helpful description of an algorithm should include three elements:

1) An informal introduction that illustrates the problem and its solution by well-chosen examples.

2) A concise definition of the general problem and an explanation of the computational theory.

¹Copyright©1992 Per Brinch Hansen

3) A well-structured complete algorithm written in a standard programming language using the same terminology as the theoretical discussion.

The best textbooks on algorithms satisfy all three requirements. However, computer scientists do not always appreciate that a professional programmer cannot rely on a complex algorithm unless it is written in a programming language that runs on an available computer. There are too many pitfalls in purely theoretical arguments about “pseudocode” that cannot be compiled and executed on any computer.

We are aware of only four texts that deal with multiple-length division. We will briefly discuss how well they satisfy the above criteria.

In 1969 *Donald Knuth* published a volume on *Seminumerical Algorithms* [5]. This work contains the most comprehensive treatment of multiple-length arithmetic:

1) In his introduction to long division, Knuth gives a three-line hint of an example and remarks that “It is clear that this same idea works in general.”

2) Knuth covers most of the relevant theory from the literature.

3) Knuth presents a division algorithm in three different ways. The first version is a mixture of English, mathematical notation, and *goto* statements. The second one is an informal flowchart that merely shows the flow of control. The third attempt is written in the assembly language MIX with informal comments. Each of these versions conveys insight, but none of them inspire complete confidence in the finer details. They are obviously not well-structured algorithms written in a standard programming language.

However, we must keep in mind that Knuth wrote the first edition of his book in the late 1960s, when structured programming was still in its infancy.

Ten years later, *Suad Alagić* and *Michael Arbib* wrote a book on *The Design of Well-Structured and Correct Programs* [6]. This text has two pages on long division:

1) Alagić and Arbib do not illustrate long division by examples.

2) They state four theorems from Knuth without motivation, proof, or reference.

3) Their algorithm for long division consists of 56 lines written in Pascal. The entire program is a single compound statement composed of shorter statements according to the principles of structured programming. However, the lack of procedures makes it hard to study the algorithm bottom-up (or top-down) at different levels of detail. The program assumes that array dimensions correspond exactly to the lengths of operands. As it stands, the program cannot be compiled and executed. (It has one syntax error and two undeclared names.) The algorithm is helpful, but not sufficiently developed for software design.

Clearly, this program could have been developed further. If the authors had finished the job, there would have been no need to write this tutorial. Alagić and Arbib make a devastating remark about this part of their book: “The reader who has struggled through [our program]—which is typical of the way in which programs are

presented and documented—may come to understand the advantages of a top-down approach to presenting a program.”

This brings us to *Derick Wood's* textbook on *Paradigms and Programming with Pascal* [7], which presents a slightly different form of long division:

- 1) Wood carefully explains long division by an example.
- 2) His method underestimates quotient digits and corrects them by a slow loop that uses multiple-length arithmetic. The main problem is the lack of a theory that predicts the maximum number of corrections required.
- 3) Wood's algorithm consists of seven procedures, three of which are left as exercises. With one exception, the remaining procedures are well-structured and well-explained. However, the main procedure for long division contains a complicated *while* statement that tests and prepares its own termination in the middle of the loop.

In his book on *Prime Numbers and Computer Methods for Factorization*, *Hans Riesel* [8] makes the following observation: “Unfortunately... the performance of exact computations on large integers has a limited appeal, and computer manufacturers do not find it profitable to include such facilities in the software that goes with their hardware. This means that the reader may have to construct such a package himself for the computer he is using... [We] shall discuss ways in which this can be done”.

- 1) Riesel has no examples of long division.
- 2) According to Riesel, “Division is by far the most complicated of the four elementary operations.” In spite of that, he immediately adds that “We shall only sketch... division”. This warning is followed by a single page of hints with no theoretical analysis.
- 3) Without further explanation, Riesel presents a Pascal procedure of 45 lines, which uses *goto* statements (instead of *while* and *for* statements) to implement iteration. Although written in a standard language, this procedure is too hard to follow.

These evaluations of existing textbooks are written not for the sake of criticizing the authors, but to explain why it is necessary to discuss a fundamental algorithm that has been known for five centuries [9]. When you need multiple-length division on a computer, you will look in vain for a textbook that combines an elegant algorithm with a simple explanation.

In the following, we illustrate the subtleties of long division by examples, define the problem concisely, summarize the theory, and develop a complete Pascal algorithm using a consistent terminology. We also derive the complexity of the algorithm and explain how the radix is selected. The appendix contains proofs of the theorems on which the algorithm is based.

LONG DIVISION

Most computers limit integer arithmetic to operands of 32–64 bits, corresponding to 8–17 decimal digits. A larger integer must be represented by an array of digits, each occupying a single machine word. The arithmetic operations on multiple-length integers are serial operations that imitate paper-and-pencil methods.

If a machine word represents a decimal digit, a 100-digit decimal number requires 100 machine words. However, if we use radix 1,000 (instead of ten), the same number occupies 34 words only. A large radix reduces both the memory space of multiple-length integers and the execution time of the serial arithmetic.

Multiple-length division is surprisingly difficult. The following example illustrates *long division* of decimal numbers, as we learned it in school.

Example 1:

$$\begin{array}{r}
 3098 \\
 102 \overline{)0316097} \\
 \underline{0306} \\
 0100 \\
 \underline{0000} \\
 1009 \\
 \underline{0918} \\
 0917 \\
 \underline{0816} \\
 101
 \end{array}$$

The initial remainder is the dividend 316097 extended with a leading zero: 0316097. (The purpose of the extra digit will soon become apparent.)

The quotient digits are computed one at a time:

(1) Since the divisor has three digits, we divide the four leading digits of the remainder by the divisor:

$$0316 \text{ div } 102 = 3$$

This gives us the leading digit of the quotient. The remainder is then reduced to 010097 as shown.

(2) We divide the four leading digits of the new remainder by the divisor to get the next quotient digit:

$$0100 \text{ div } 102 = 0$$

This leaves a remainder of 10097.

(3)–(4) We use the same method to compute the last two digits of the quotient:

$$1009 \text{ div } 102 = 9$$

$$0917 \text{ div } 102 = 8$$

The final remainder is 101.

In each step, we treat the four leading digits of the remainder r as an integer $r\{4\}$ and use the divisor d to compute a *quotient digit* q_k :

$$q_k = r\{4\} \text{ div } d$$

The integer $r\{4\}$ is called a *prefix* of the remainder.

The following table is a different representation of the division steps in Example 1.

k	$r\{4\}$	q_k
3	0316	3
2	0100	0
1	1009	9
0	0917	8

Without a zero in front of the initial remainder, the computation of the first quotient digit would be a special case:

$$316 \text{ div } 102 = 3$$

requiring three digits of the remainder (instead of four).

If the divisor has many digits, a quotient digit can seldom be computed directly, but must be estimated and corrected, if necessary. The main challenge is to replace human intuition about this process by an efficient iterative algorithm.

The three leading digits of the remainder define a shorter *prefix* $r\{3\}$. Similarly, the two leading digits of the divisor define a *prefix* $d\{2\}$. We will use

$$r\{3\} \text{ div } d\{2\}$$

as an *initial estimate* of the quotient digit q_k . Since a decimal digit must be less than 10, the initial estimate q_e is defined as follows:

$$q_e = \min(r\{3\} \text{ div } d\{2\}, 9)$$

The *error* of the initial estimate is the difference:

$$\Delta q = q_e - q_k$$

In Example 1, where $d\{2\} = 10$, the initial estimate produces the following sequence of digits:

k	$r\{4\}$	q_k	$r\{3\}$	q_e	Δq
3	0316	3	031	3	0
2	0100	0	010	1	1
1	1009	9	100	9	0
0	0917	8	091	9	1

In every step,

$$0 \leq \Delta q \leq 1$$

Later we will see that this inequality always holds. So, the most obvious idea is to correct a *trial digit* q_t as follows:

```
qt := qe;
if r{4} < d*qt then qt := qt - 1
```

At the end of this iteration, $q_t = q_k$.

In Example 1, half of the initial guesses require a single correction. The number of corrections can be reduced dramatically by *scaling* the operands before the division: We multiply the divisor and the dividend by the same digit. The scaling must make the leading digit of the divisor at least equal to half of the radix without changing the length of the divisor. This is called *normalization*.

Example 2:

Example 1 is normalized by multiplying both operands by five:

$$\begin{aligned} 102 * 5 &= 510 \\ 0316097 * 5 &= 1580485 \end{aligned}$$

The leading digit of the normalized divisor is equal to 5, which is half of the radix 10. The normalized divisor still has three digits only. Since the original dividend already had an extra digit-position, normalization does not change its length either. This is another reason for adding the extra digit.

After normalization, division proceeds as usual:

$$\begin{array}{r} 3098 \\ 510 \overline{)1580485} \\ \underline{1530} \\ 0504 \\ \underline{0000} \\ 5048 \\ \underline{4590} \\ 4585 \\ \underline{4080} \\ 505 \end{array}$$

Finally, the remainder is divided by the scaling factor to obtain the remainder of the original problem:

$$505 \text{ div } 5 = 101$$

In each step, the initial estimate q_e is the prefix $r\{3\}$ divided by the prefix $d\{2\} = 51$:

k	$r\{4\}$	q_k	$r\{3\}$	q_e	Δq
3	1580	3	158	3	0
2	0504	0	050	0	0
1	5048	9	504	9	0
0	4585	8	458	8	0

In this example, the initial estimates are exact estimates of the corresponding quotient digits. However, in general, normalized division occasionally requires correction of an initial estimate. In decimal arithmetic, the average number of corrections is less than 20%. For higher radices, corrections are rarely necessary.

After this intuitive exploration, we are ready for a concise analysis of the problem.

THE ESSENCE OF THE PROBLEM

We are considering long division of two natural numbers, x and y :

$$q = x \text{ div } y \tag{1}$$

$$r = x \text{ mod } y$$

where $x \geq 0$ and $y > 0$.

The quotient q and the remainder r are natural numbers which satisfy the constraints:

$$x = y * q + r \tag{2}$$

$$0 \leq r \leq y - 1 \tag{3}$$

Each number is represented by an array of digits in *radix* b . The *dividend* x has n digits:

$$x = x_{n-1}b^{n-1} + x_{n-2}b^{n-2} + \dots + x_0 \tag{4}$$

while the *divisor* y has m digits:

$$y = y_{m-1}b^{m-1} + y_{m-2}b^{m-2} + \dots + y_0 \tag{5}$$

Two special cases immediately arise:

(1) $m = 1$: If the divisor is a single-digit number, we will use a very simple division algorithm. Since zero is a single-digit number, this algorithm will also detect overflow.

(2) $m > n$: If the divisor is longer than the dividend, the quotient is zero and the remainder is x .

In the theoretical analysis, we will concentrate on the remaining case

$$2 \leq m \leq n \tag{6}$$

where the divisor has at least two digits, and the dividend has at least as many digits as the divisor.

Before the division, we multiply the operands by a *scaling factor* $f \geq 1$ that will be defined later. The following abstract program defines *normalized division*:

```

var x, y, q, r, d, f: integer;
begin
    r := x*f;
    d := y*f;
    q := r div d;
    r := (r mod d) div f
end

```

The *normalized divisor* $d = y * f$ has m digits:

$$d = d_{m-1}b^{m-1} + d_{m-2}b^{m-2} + \dots + d_0 \quad (7)$$

Since $y > 0$ and $f \geq 1$, we also have $d > 0$.

The *quotient* q has $n - m + 1$ digits:

$$q = q_{n-m}b^{n-m} + q_{n-m-1}b^{n-m-1} + \dots + q_0 \quad (8)$$

The initial *remainder* $r = x * f$ has $n + 1$ digits. Immediately before the computation of quotient digit q_k , the remainder r has been reduced to $k + m + 1$ digits:

$$r = r_{k+m}b^{k+m} + r_{k+m-1}b^{k+m-1} + \dots + r_0 \quad (9)$$

where $0 \leq k \leq n - m$.

The leading $m + 1$ digits of the remainder define a *prefix* $r\{m + 1\}$:

$$r\{m + 1\} = r_{k+m}b^m + r_{k+m-1}b^{m-1} + \dots + r_k \quad (10)$$

The essence of multiple-length division is the computation of a single *quotient digit* q_k :

$$q_k = r\{m + 1\} \mathbf{div} d \quad (11)$$

by iteration.

To simplify the algebra a bit, we will assume that the *radix* b is *even*, say, a power of two or ten:

$$b \mathbf{div} 2 = b/2 \geq 1 \quad (12)$$

TRIAL ITERATION

The initial estimate

The computation of a quotient digit q_k is an iteration that decrements an initial estimate until it equals q_k . The most conservative guess is $b - 1$, which requires $O(b)$ corrections. Fortunately, there is a much better choice.

The three leading digits of the remainder r define a *prefix* $r\{3\}$:

$$r\{3\} = r_{k+m}b^2 + r_{k+m-1}b + r_{k+m-2} \quad (13)$$

The *prefix* $d\{2\}$ consists of the two leading digits of the divisor d :

$$d\{2\} = d_{m-1}b + d_{m-2} \quad (14)$$

where

$$2 \leq m \leq k + m \leq n \quad (15)$$

We assume that $r\{3\}$ and $d\{2\}$ can be represented as single-length integers.

We will use

$$q_e = \min(r\{3\} \text{ div } d\{2\}, b - 1) \quad (16)$$

as the *initial estimate* of q_k . Obviously, $0 \leq q_e \leq b - 1$.

The following theorems show that q_e is an excellent guess. They are due to E. V. Krishnamurthy and Salil Nandi [10]. (The appendix includes the proofs of all theorems used.)

Theorem 1: $q_k \leq q_e$

Theorem 2: $q_e \leq q_k + 1$

By combining Theorems 1 and 2 with the assumption that q_k is a digit, we obtain the inequality:

$$0 \leq q_k \leq q_e \leq q_k + 1 \leq b \quad (17)$$

So, the initial estimate q_e is either correct or off by 1.

Trial correction

The computation of a *quotient digit* q_k is based on two simple theorems about any *trial digit* q_t :

Theorem 3: If $r\{m+1\} < d * q_t$ then $q_k < q_t$

Theorem 4: If $r\{m+1\} \geq d * q_t$ then $q_t \leq q_k$

These theorems and (17) suggest the following *trial iteration*:

```
qt := qe;
if r{m+1} < d*qt then qt := qt - 1
```

Here is the same algorithm with assertions added:

```

{0 ≤ qk ≤ qe ≤ qk + 1 ≤ b by (17)}
qt := qe;
{0 ≤ qk ≤ qt ≤ qk + 1 ≤ b}
if r{m+1} < d*qt then
    {0 ≤ qk < qt ≤ qk + 1 ≤ b by Theorem 3}
    qt := qt - 1
    {0 ≤ qk ≤ qt < qk + 1 ≤ b}
else {0 ≤ qk ≤ qt ≤ qk < b by Theorem 4}
    
```

The *if* statement terminates with the postcondition:

$$0 \leq q_k \leq q_t \leq q_k \leq b - 1$$

which implies that $q_t = q_k$.

We will show that the frequency of trial corrections depends on the *leading digit* $d\{1\}$ of the divisor d :

$$d\{1\} = d_{m-1} \tag{18}$$

where $1 \leq d\{1\} \leq b - 1$.

The computation of quotient digit q_k is equivalent to the integer division

$$q_k = r\{m + 1\} \text{ div } d$$

which leaves a remainder

$$r\{m + 1\} \bmod d = r\{m + 1\} - d * q_k$$

where $0 \leq r\{m + 1\} \bmod d < d$.

We assume that the relative remainder

$$z_k = (r\{m + 1\} \bmod d) / d = r\{m + 1\} / d - q_k$$

is a random variable with a uniform distribution between 0 and 1.

At this point, we need another theorem:

Theorem 5:

If $q_k = q_e - 1$ then $r\{m + 1\} / d - q_k > 1 - 1/d\{1\}$

The probability that a correction is required is

$$\mathbf{P}(q_k = q_e - 1) = \mathbf{P}(z_k > 1 - 1/d\{1\})$$

In other words,

$$\mathbf{P}(q_k = q_e - 1) < 1/d\{1\} \tag{19}$$

To reduce the number of corrections, we should obviously make the leading digit of the divisor as large as possible. This requires normalization of the operands.

Normalization

The divisor d is *normalized* if its leading digit is at least half of the radix b :

$$1 \leq b/2 \leq d\{1\} \leq b - 1 \tag{20}$$

This requirement implies that

$$1 \leq b^m/2 \leq d \leq b^m - 1 \tag{21}$$

The *leading digit* $y\{1\}$ of the original divisor y is

$$y\{1\} = y_{m-1} \tag{22}$$

where $1 \leq y\{1\} \leq b - 1$. This digit determines the *scaling factor* f :

$$f = b \text{ div } (y\{1\} + 1) \tag{23}$$

Obviously, $1 \leq f \leq b/2$.

The following theorem shows that normalization is done correctly.

Theorem 6: $b^m/2 \leq y * f \leq b^m - 1$

From now on, we assume that the divisor $d = y * f$ is normalized. By (19) and (20), the probability that an initial estimate q_e needs to be decremented is bounded as follows:

$$\mathbf{P}(q_k = q_e - 1) < 2/b$$

In decimal division, at least 80% of the initial estimates are correct. For radix 1,000, each guess is correct with probability 0.998.

PASCAL ALGORITHM

With this background, we are ready to write a Pascal algorithm for multiple-length division. The theoretical discussion introduced names for the most important constants and variables. The algorithm uses exactly the same terminology and is presented in bottom-up form. (If you prefer top-down design, please read the rest of this section backwards.)

Number representation

A natural *number* x is represented by an array of $w + 1$ digits in *radix* b , say,

```

const b = 1000; w = 100;
type number = array [0..w] of integer;
var x: number;
    
```

The algorithm can divide natural numbers with 1 to w digits. During division, the remainder is extended with an additional position.

Every *digit* $x[k]$ is an integer in the range:

$$0 \leq x[k] \leq b - 1 \quad \text{for } 0 \leq k \leq w$$

The k^{th} digit of x represents the integer $x[k] * b^k$. We will use the words *left* and *right* to refer to the high-order and low-order positions of a number.

The number *zero* is represented by a global variable with $w + 1$ zeros

var zero: number

The *length* of a natural number is the number of significant digits in the corresponding array. Algorithm 1 determines the length of a number x by a linear search from left to right. The search stops at the first nonzero digit or at the rightmost digit, whichever is reached first. This variant of linear searching was derived in [11].

```

function length(x: number)
  : integer;
var i, j: integer;
begin
  i := w; j := 0;
  while i <> j do
    if x[i] <> 0 then j := i
    else i := i - 1;
  length := i + 1
end

```

Algorithm 1

Partial arithmetic

The simplest part of multiple-length division is the multiplication or division of a natural number by a single digit. To avoid confusing these *partial operations* with complete multiple-length operations, they are called *product*, *quotient*, and *remainder* (instead of multiply, divide, and modulo).

<i>Procedure:</i>	<i>Operation:</i>
product(x,y,k)	$x := y * k$
quotient(x,y,k)	$x := y \text{ div } k$
remainder(x,y,k)	$x := y \text{ mod } k$

Each operation involves two natural numbers x and y , and a digit k . The procedures are straightforward implementations of familiar paper-and-pencil methods.

A partial *product* is computed, digit by digit, from right to left using a *carry* (Algorithm 2). If *overflow* occurs, the execution halts.

```

procedure product(var x: number;
  y: number; k: integer);
var carry, i, m, temp: integer;
begin
  m := length(y);
  x := zero; carry := 0;
  for i := 0 to m - 1 do
    begin
      temp := y[i]*k + carry;
      x[i] := temp mod b;
      carry := temp div b
    end;
  if m <= w then x[m] := carry
  else if carry <> 0 then overflow
end

```

Algorithm 2

3). The *quotient* of a partial division is calculated serially from left to right (Algorithm

```

procedure quotient(var x: number;
  y: number; k: integer);
var carry, i, m, temp: integer;
begin
  m := length(y);
  x := zero; carry := 0;
  for i := m - 1 downto 0 do
    begin
      temp := carry*b + y[i];
      x[i] := temp div k;
      carry := temp mod k
    end
  end

```

Algorithm 3

4). The *remainder* of a partial division is the last carry (compare Algorithms 3 and

```

procedure remainder(var x: number;
  y: number; k: integer);
var carry, i, m: integer;
begin
  m := length(y);
  x := zero; carry := 0;
  for i := m - 1 downto 0 do
    carry := (carry*b + y[i]) mod k;
  x[0] := carry
end

```

Algorithm 4

Prefix arithmetic

The computation of a quotient digit $q[k]$ breaks down into simpler *prefix operations*. In the following, $x[i..j]$ denotes digits i through j of a natural number x .

The assignment

$$q_t := \text{trial}(r, d, k, m)$$

defines a *trial digit* $q_t = qe$, which is an initial estimate of q_k . The operands of the *trial function* are prefixes of the remainder r and the divisor d :

$$r\{3\} = r[k + m..k + m - 2] \quad d\{2\} = d[m - 1..m - 2]$$

where

$$2 \leq m \leq k + m \leq w \tag{24}$$

The initial estimate is computed as described earlier (Algorithm 5). Strictly speaking, the trial function should verify that its *precondition* holds. However, since (24) turns out to be an invariant of long division, this assumption is described by a comment only.

```

function trial(r, d: number;
  k, m: integer): integer;
var d2, km, r3: integer;
begin
  {2 <= m <= k+m <= w}
  km := k + m;
  r3 := (r[km]*b + r[km - 1])*b
    + r[km - 2];
  d2 := d[m - 1]*b + d[m - 2];
  trial := min(r3 div d2, b - 1)
end

```

Algorithm 5

Two procedures define prefix comparison and subtraction:

<i>Procedure:</i>	<i>Operation:</i>
<code>smaller(r, dq, k, m)</code>	$r\{m+1\} < dq$
<code>difference(r, dq, k, m)</code>	$r\{m+1\} := r\{m+1\} - dq$

The $(m + 1)$ -place operands are

$$r\{m + 1\} = r[k + m..k] \quad d * q_t = dq[m..0]$$

where

$$0 \leq k \leq k + m \leq w \tag{25}$$

These operations are equivalent to

$$r < dq * b^k \quad r := r - dq * b^k$$

Since the division invariant (24) implies the precondition (25), these procedures do not verify this assumption.

The boolean function that determines if the prefix $r\{m + 1\}$ is *smaller* than the product dq is a variant of the *length* function (see Algorithms 1 and 6).

```

function smaller(r, dq: number;
  k, m: integer): boolean;
var i, j: integer;
begin
  {0 <= k <= k+m <= w}
  i := m; j := 0;
  while i <> j do
    if r[i + k] <> dq[i]
      then j := i
      else i := i - 1;
  smaller := r[i + k] < dq[i]
end

```

Algorithm 6

The subtraction of the product dq from the $m+1$ high-order digits of the remainder r proceeds from right to left using a *borrow*. If the *difference* is negative, the execution halts (Algorithm 7).

```

procedure difference(var r: number;
    dq: number; k, m: integer);
var borrow, diff, i: integer;
begin
    {0 <= k <= k+m <= w}
    borrow := 0;
    for i := 0 to m do
        begin
            diff := r[i + k] - dq[i]
                - borrow + b;
            r[i + k] := diff mod b;
            borrow := 1 - diff div b
        end;
    if borrow <> 0 then overflow
end

```

Algorithm 7

Division algorithms

Algorithm 8 defines *long division*

$$q := x \text{ div } y; r := x \text{ mod } y$$

of two natural numbers

$$x[n - 1..0] \quad y[m - 1..0]$$

where

$$2 \leq m \leq n \leq w \tag{26}$$

After normalization of the operands, the *quotient* q is computed, digit by digit, from left to right. The quotient iteration maintains the invariant

$$2 \leq m \leq k + m \leq n \leq w \tag{27}$$

The final *remainder* r is divided by the *scaling factor* f .

```

procedure longdivide(x, y: number;
  var q, r: number; n, m: integer);
var d, dq: number; f, k, qt: integer;
begin
  {2 <= m <= n <= w}
  f := b div (y[m - 1] + 1);
  product(r, x, f);
  product(d, y, f);
  q := zero;
  for k := n - m downto 0 do
    begin
      {2 <= m <= k+m <= n <= w}
      qt := trial(r, d, k, m);
      product(dq, d, qt);
      if smaller(r, dq, k, m) then
        begin
          qt := qt - 1;
          product(dq, d, qt)
        end;
      q[k] := qt;
      difference(r, dq, k, m)
    end;
  quotient(r, r, f)
end

```

Algorithm 8

The complete algorithm for *multiple-length division* uses simpler methods for the special cases, where the *divisor* y has a single-digit only or is longer than the *dividend* x (Algorithm 9). In all other cases, the algorithm uses long division after establishing the precondition (26).

```

procedure division(x, y: number;
    var q, r: number);
var m, n, y1: integer;
begin
    m := length(y);
    if m = 1 then
        begin
            y1 := y[m - 1];
            if y1 > 0 then
                begin
                    quotient(q, x, y1);
                    remainder(r, x, y1)
                end
            else overflow
        end
    else
        begin
            n := length(x);
            if m > n then
                begin
                    q := zero; r := x
                end
            else {2 <= m <= n <= w}
                longdivide(x, y, q, r, n, m)
            end
        end
    end

```

Algorithm 9

Complexity

The complexity of long division is determined by the $n - m + 1$ quotient steps. Each step is dominated by product and difference operations on $(m + 1)$ -place operands. Consequently, the complexity is

$$O((n - m + 1)(m + 1))$$

For random divisors with uniformly distributed lengths between 2 and n , the average run time T is close to

$$T \approx \frac{1}{n} \int_2^n (n - m + 1)(m + 1) dm = O(n^2) \quad \text{for } n \gg 2$$

Since

$$10^N \approx b^n \quad \text{for } n \approx N / \log b$$

a decimal number with N digits corresponds to a radix- b number with $N/\log b$ digits. Consequently,

$$T \approx O\left((N/\log b)^2\right)$$

Using radix 1,000, instead of ten, reduces the average division time by a factor of 9. Radix 10,000 makes multiple-length division 16 times faster than decimal division.

Radix selection

Algorithms 3 and 4 include integer expressions of the form

$$d * b + d$$

where the d 's denote arbitrary digits. The corresponding values can be represented by standard integers if we chose a radix b that satisfies the inequality

$$d * b + d \leq b^2 - 1 \leq \text{maxint}$$

where *maxint* is the largest standard integer.

For 32-bit integers, this means

$$b^2 \leq 2^{31}$$

or $b \leq 46,340$. A possible choice is $b = 10,000$.

However, the *trial function* (Algorithm 5) has an expression of the form

$$(d * b + d) * b + d$$

The corresponding inequality is

$$(d * b + d) * b + d \leq b^3 - 1 \leq \text{maxint}$$

For 32-bit integers, we must therefore use a smaller radix, for which

$$b^3 \leq 2^{31}$$

that is, $b \leq 1,290$. For example, $b = 1,000$.

If the available computer supports both 32 and 64-bit integers, a radix of 10,000 is feasible if we use *double-precision arithmetic* in the trial function.

FINAL REMARKS

We have developed a Pascal algorithm for long division of natural numbers and have explained the theory behind the algorithm. This paper is merely an interpretation and formalization of the original ideas of the authors referenced in the text. It was written only because we were unable to find an elegant, complete algorithm described with convincing clarity in the literature.

ACKNOWLEDGEMENTS

I thank Konrad Jahn who taught me how to divide in the fourth grade in Denmark. I am indebted to Jonathan Greenfield who rediscovered Theorem 2 and brought it to my attention. It is also a pleasure to acknowledge the comments of Suad Alagić, Christian Gram, Hans Riesel, and Peter Villemoes.

APPENDIX

The appendix summarizes the laws of prefixes and integer division and proves the theorems about long division.

Prefix laws

By (5), (22):

$$y\{1\}b^{m-1} \leq y < (y\{1\} + 1)b^{m-1} \tag{28}$$

By (7), (18):

$$d\{1\}b^{m-1} \leq d < (d\{1\} + 1)b^{m-1} \tag{29}$$

By (7), (14):

$$d\{2\}b^{m-2} \leq d < (d\{2\} + 1)b^{m-2} \tag{30}$$

By (10), (13):

$$r\{3\}b^{m-2} \leq r\{m+1\} < (r\{3\} + 1)b^{m-2} \tag{31}$$

Division laws

Any integer division $q = x \text{ div } y$ satisfies three equivalent laws:

$$x/y - 1 < q \leq x/y \tag{32}$$

$$x < y(q + 1) \leq x + y \tag{33}$$

$$x - y + 1 \leq y * q \leq x \tag{34}$$

The following instances of these laws apply to the integer divisions that define the quotients f , q_e and q_k , where

$$f = b \operatorname{div} (y\{1\} + 1)$$

$$q_e = \min(r\{3\} \operatorname{div} d\{2\}, b - 1)$$

$$q_k = r\{m + 1\} \operatorname{div} d$$

By (32):

$$b/(y\{1\} + 1) - 1 < f \leq b/(y\{1\} + 1) \quad (35)$$

If $q_e = r\{3\} \operatorname{div} d\{2\}$ then by (34):

$$r\{3\} - d\{2\} + 1 \leq d\{2\} * q_e \leq r\{3\} \quad (36)$$

By (34):

$$r\{m + 1\} - d + 1 \leq d * q_k \leq r\{m + 1\} \quad (37)$$

By (33):

$$r\{m + 1\} < d(q_k + 1) \leq r\{m + 1\} + d \quad (38)$$

Proof of theorem 1

Theorem: $q_k \leq q_e$

Proof:

1) If $q_e = r\{3\} \operatorname{div} d\{2\}$ then

$$d(q_k - q_e) \leq r\{m + 1\} - d\{2\}q_e b^{m-2} \quad \text{by (30), (37)}$$

$$< ((r\{3\} + 1) - (r\{3\} - d\{2\} + 1))b^{m-2} \quad \text{by (31), (36)}$$

$$= d\{2\}b^{m-2}$$

$$\leq d \quad \text{by (30)}$$

Since $d > 0$, we have $q_k - q_e < 1$, which implies $q_k - q_e \leq 0$.

2) If $q_e = b - 1$ then

$$q_k \leq q_e \quad \text{by } q_k \leq b - 1$$

Proof of theorem 2

Theorem: $q_e \leq q_k + 1$

Proof by contradiction: If $q_e \geq q_k + 2$ then

$$\begin{aligned}
 r\{m+1\} &< d(q_k+1) && \text{by (38)} \\
 &\leq d(q_e-1) && \text{by } q_e \geq q_k + 2 \\
 &< (d\{2\} + 1)q_e b^{m-2} - d && \text{by (30)} \\
 &\leq (r\{3\} + q_e)b^{m-2} - d && \text{by (36)} \\
 &< (r\{3\} + b)b^{m-2} - d && \text{by } q_e < b \\
 &\leq r\{3\}b^{m-2} && \text{by } d \geq b^{m-1} \\
 &\leq r\{m+1\} && \text{by (31)}
 \end{aligned}$$

that is, $r\{m+1\} < r\{m+1\}$, a contradiction.

Proof of theorem 3

Theorem: If $r\{m+1\} < d * q_t$ then $q_k < q_t$

Proof:

$$\begin{aligned}
 d(q_k - q_t) &\leq r\{m+1\} - d * q_t && \text{by (37)} \\
 &< 0 && \text{by } r\{m+1\} < d * q_t
 \end{aligned}$$

Since $d > 0$, we have $q_k - q_t > 0$.

Proof of theorem 4

Theorem: If $r\{m+1\} \geq d * q_k$ then $q_t \leq q_k$

Proof by contradiction: If $q_t \geq q_k + 1$ then

$$\begin{aligned}
 r\{m+1\} &< d(q_k+1) && \text{by (38)} \\
 &\leq d * q_t && \text{by } q_t \geq q_k + 1 \\
 &\leq r\{m+1\} && \text{by } r\{m+1\} \geq d * q_t
 \end{aligned}$$

that is, $r\{m+1\} < r\{m+1\}$, a contradiction.

Proof of theorem 5

Theorem: If $q_k = q_e - 1$ then $r\{m+1\}/d - q_k > 1 - 1/d\{1\}$

Proof:

$$\begin{aligned}
 r\{m+1\} - d * q_k &= d + r\{m+1\} - d * q_e && \text{by } q_k = q_e - 1 \\
 &> d + (r\{3\} - (d\{2\} + 1)q_e)b^{m-2} && \text{by (30), (31)} \\
 &\geq d - q_e b^{m-2} && \text{by (36)} \\
 &> d - b^{m-1} && \text{by } q_e < b \\
 &= d(1 - b^{m-1}/d) && \text{by } d > 0 \\
 &\geq d(1 - 1/d\{1\}) && \text{by (29)}
 \end{aligned}$$

Consequently, $r\{m+1\}/d - q_k > 1 - 1/d\{1\}$.

Proof of theorem 6

Theorem: $b^m/2 \leq y * f \leq b^m - 1$

1) Upper bound:

$$\begin{aligned}
 y * f &\leq ((y\{1\} + 1)b^{m-1} - 1)f && \text{by (28)} \\
 &\leq (y\{1\} + 1)f * b^{m-1} - 1 && \text{by } f \geq 1 \\
 &\leq b^m - 1 && \text{if } f \leq b/(y\{1\} + 1)
 \end{aligned}$$

The largest possible scaling factor is

$$f = b \operatorname{div} (y\{1\} + 1)$$

2) Lower bound:

If $y\{1\} < b/2$ then

$$\begin{aligned}
 y\{1\}f &> y\{1\}(b/(y\{1\} + 1) - 1) && \text{by (35)} \\
 &= (b/2 - y\{1\} - 1)(y\{1\} - 1)/(y\{1\} + 1) + b/2 - 1 \\
 &\geq b/2 - 1 && \text{by } 1 \leq y\{1\} \leq b/2 - 1
 \end{aligned}$$

Since $y\{1\}f > b/2 - 1$ and $b/2$ is an integer, we have $y\{1\}f \geq b/2$. Consequently,

$$\begin{aligned}
 y * f &\geq y\{1\}f * b^{m-1} && \text{by (28)} \\
 &\geq b^m/2 && \text{by } y\{1\}f \geq b/2
 \end{aligned}$$

If $y\{1\} \geq b/2$ then

$$\begin{aligned} y * f &\geq y\{1\}b^{m-1} && \text{by } f \geq 1, (28) \\ &\geq b^m/2 && \text{by } y\{1\} \geq b/2 \end{aligned}$$

REFERENCES

1. P. Brinch Hansen and R. House, 'The Cobol compiler for the Siemens 3003', *BIT*, **6**, 1–23 (1966).
2. R. L. Rivest, A. Shamir and L. M. Adleman, 'A method for obtaining digital signatures for public-key cryptography', *Communications of the ACM*, **21**, 120–126 (1978).
3. M. O. Rabin, 'Probabilistic algorithms for testing primality', *Journal of Number Theory*, **12**, 128–138 (1980).
4. P. Brinch Hansen, 'Primality testing', School of Computer and Information Science, Syracuse University, Syracuse, NY, 1992.
5. D. Knuth, *The Art of Computer Programming. Volume 2: Seminumerical Algorithms*. Addison-Wesley, Reading, MA, 1969.
6. S. A. Alagić and M. A. Arbib, *The Design of Well-Structured and Correct Programs*. Springer-Verlag, New York, 1978.
7. D. Wood, *Paradigms and Programming with Pascal*. Computer Science Press, Rockville, MD, 1984.
8. H. Riesel, *Prime Numbers and Computer Methods for Factorization*. Birkhäuser, Boston, MA, 1985.
9. D. E. Smith and J. Ginsburg, 'From numbers to numerals and from numerals to computation', in J. R. Newman (ed.), *The World of Mathematics*. Volume 1. Tempus, Redmond, WA, 433–454, 1988.
10. E. V. Krishnamurthy and S. K. Nandi, 'On the normalization requirement of divisor in divide-and-correct methods', *Communications of the ACM*, **12**, 809–813 (1967).
11. P. Brinch Hansen, 'The linear search rediscovered', *Structured Programming*, **11**, 53–55 (1990).